
1.0 ABSTRACT  

Time-to-event endpoints such as overall survival are commonly used as outcomes in oncology clinical trials, and one 
of the best graphical displays of these outcomes is the Kaplan-Meier curve.  The macro NEWSURV, which has been 
presented at SAS® conferences in 2014 and 2015, has become popular as a powerful tool for creating highly 
customizable publication quality Kaplan-Meier plots.  The positive feedback for the NEWSURV macro has led to the 
program being updated with additional features and improvements, and has led to additional macros being created 
for the situations that NEWSURV does not cover.  A pair of macros, NEWSURV_ADJ_INVWTS and 
NEWSURV_ADJ_DIRECT, create adjusted Kaplan-Meier curves using two specific methods: inverse weights and 
direct adjustment.  The macro NEWSURV_DATA allows the user to provide their own dataset with pre-calculated 
survival function estimates that the program will then use to create a high quality plot in the NEWSURV style.  This 
opens up the flexibility for any adjusted or weighted curve to be plotted at publication quality.  The following is a paper 
describing the updated functionality of NEWSURV and an overview on the methods used with the new expansion 
macros.    

2.0 INTRODUCTION  

Time-to-event endpoints such as overall survival are commonly used as outcomes in oncology clinical trials, and one 
of the best graphical displays of these outcomes is the Kaplan-Meier curve.  The LIFTETEST procedure automatically 
creates great Kaplan-Meier curves, but has two shortcomings that led to the creation of the NEWSURV macro.  The 
first is the difficulty of customizing the plot colors, labels, axes, titles, and other attributes.  The second is that for 
publications statistics such as patient counts, median time-to-event, and p-values are commonly displayed within the 
graph, and this is not possible strictly through the LIFETEST procedure.  The NEWSURV macro was designed to 
create highly customizable Kaplan-Meier curves that could also display several commonly used statistics 
automatically within the graph.  The NEWSURV macro was presented at multiple SAS conferences in 2014

1,2
 and 

2015
3
, and has since evolved into a more powerful tool due to positive feedback and requests for improvements.  The 

NEWSURV macro has added the capability to plot cumulative incidence and perform competing risk analysis using 
the Fine and Gray methodology.  The macro can now also add confidence intervals and reference lines to the graph 
to make comparing groups easier.   

The most common improvement request was to add the ability to create adjusted Kaplan-Meier curves, and this 
turned out to be by far the most difficult.  There is not a single method to adjust Kaplan-Meier curves as the 
appropriate method strongly depends on the type of data being analyzed.  The different methodologies are explained 
within a paper by Therneau, Crowson, and Atkinson

4
.  Discussions with several Mayo Clinic statisticians led to three 

spin-off macros to be designed instead of directly incorporating adjusted survival curves into NEWSURV itself.  This 
was to not give the inappropriate message that the methods included within NEWSURV are the “correct” methods.  
The inverse weights methodology is used within the NEWSURV_ADJ_INVWTS macro.  The direct adjustment 
methodology is used within the NEWSURV_ADJ_DIRECT macro.  The NEWSURV_DATA macro was designed to 
allow the user to create survival function estimates that the macro will then create a high quality plot from.  This 
opens up the flexibility to use any adjustment method and still have a high quality customizable graph.   

The following paper gives a quick introduction to the NEWSURV macro, a walkthrough of the new features, and then 
a description of the three new macros with examples.  This paper will not describe the code used to create the plots 
with the Graph Template Language as this has been described thoroughly in the prior papers

1,2,3
.  These macros are 

available online using the link in section 10.0.  This link includes downloads of prior papers, presentations, macro 
versions, and a tutorial on how to use the macros. 

3.0 SAMPLE DATASETS USED IN EXAMPLES 

All of the examples in this paper use the BMT data set, the code found in the SAS auto call macro CIF.  The data set 
contains survival data on bone marrow transplant patients and has four variables: DIAGNOSIS, GENDER, FTIME, 
and STATUS. The DIAGNOSIS variable is a discrete categorical variable containing three different disease groups: 
1=ALL, 2=AML-Low Risk, and 3=AML-High Risk. The GENDER variable has two values: 1=Male and 2=Female.  The 
FTIME variable is the time from transplant, and the STATUS variable is the survival status (0=Alive, 1=Dead in 
remission, 2=relapse).  This is not the same BMT dataset that is found within the standard SASHELP library.  This 
dataset is chosen as it not only contains survival outcome data but also contains a gender variable as well for use as 

an adjusting covariate.  The status variable also has more than one event code which allows the dataset to be used 
in cumulative incidence examples.  The code used to create this dataset will be provided in appendix one. 

Table 1 shows the first several rows of the BMT data set. 

DIAGNOSIS Ftime Status Gender

ALL 2081 0 Male

ALL 1602 0 Male

ALL 1496 0 Male

ALL 1462 0 Female

ALL 1433 0 Male

Table 1. The DIAGNOSIS variable is a discrete categorical numeric variable with three levels: 1=ALL, 2=AML-
Low Risk, and 3=AML-High Risk.  The Gender variable is a discrete categorical numeric variable with two 
levels: 0=Female and 1=Male.  The FTIME variable is a numeric variable representing time from transplant.  
The STATUS variable is a numeric variable representing survival status where 0=censored, 1=death in 
remission, and 2=relapse. 

4.0 NEWSURV MACRO UPDATE 

The original NEWSURV macro was last presented at the 2015 SAS Global Forum
3
 and has received a large amount

of positive feedback and share requests both internally and externally to the Mayo Clinic.  Due to the popularity the 
macro is continuously being improved with error corrections and new features.  This section will be for giving a quick 
description of the original features of the macro for those unfamiliar as well as describing the new functionality added 
since the 2015 SAS Global Forum. 

4.1 NEWSURV MACRO PREVIOUS FEATURES 

4.1.1 Highly Customizable Survival Plot 

The purpose for creating the NEWSURV macro was to be able to create a highly customizable Kaplan-Meier curve 
using the newer Stat/Graph (SG) graphics designed by SAS in version 9.2.  The Mayo Clinic had a previous macro 
that used the older technology to create a curve called the SURV macro, so when creating the new version the name 
NEWSURV was used.  The NEWSURV macro has over 100 macro parameters that can be used to fine-tune a graph, 
but is designed such that only three are required to create an initial plot.  The rest of the parameters are setup to be 
used as needed to customize the graph to a specific project. 

Figure 1is the figure creating using only the required macro parameters. 

Figure 1. The only required macro parameters are DATA to specify a dataset, TIME to specify a time variable, 
and CENS to specify the time-to-event outcome status (Alive/dead for example).  Note that some options 
were used to change file type for the final paper. 

%newsurv(data=bmt,time=ftime,cens=status) 

Figure 2 is the figure created using optional macro parameters. 

 

Figure 2. This example uses options to split the curve into the Diagnosis groups, change line pattern and 
colors, add patients at risk, change the x-axis units, change the axes labels, and add a time-point event-free 
rate to the summary. 

%newsurv( 

    /*Required Parameters*/ 

    data=bmt,time=ftime,cens=status, 

    /*Class Variable Options*/ 

    class=diagnosis, 

    /*Line/Symbol Options*/ 

    linesize=2pt,symbolsize=10pt,color=black blue red,pattern=1, 

    /*Axis Options*/ 

    xdivisor=30.44,ylabel=Percent Alive,xlabel=Months,xmax=100,xincrement=20, 

    /*Patients-at-Risk Options*/ 

    parheader=No at Risk,paralign=labels,risklist=0 to 100 by 20, 

risklocation=bottom,risklabellocation=left,riskcolor=1, 

    /*Event-free Rate Options*/ 

    timelist=6,listtimepoints=0,kmestheader=6 month Survival, 

    /*Output File Options*/ 

    plotname=figure2); 

4.1.2 Plot Summary Table 

The second purpose of creating the NEWSURV macro was to automatically create a summary table of statistics 
within the plot itself.  Statistics such as p-values and median time-to-event are often displayed within the plot for 
publications, and having to manually create them within text-boxes was inefficient.  The NEWSURV macro allows the 
following statistics to be displayed within the plot: 

54 35 24 11 2 0AML low risk-

 Patient sample size and number of events 

 Median time-to-event 

 Hazard ratios (with specifiable reference group) 

 One or more time-point event-free rates 

 Cox model p-values 

 Logrank or Wilcoxon p-values 

 Multivariate patient counts, events, hazard ratios, and p-values 

These statistics are controlled by the DISPLAY parameter.  Only the statistics listed in the parameter list will be 
displayed, and they will be displayed in the order that they are listed. 

4.1.3 Report Table 

The NEWSURV macro also creates publication quality report tables in addition to high quality plots.  One or more 
models can be included in the report table, and one or macro calls can add to the same report table.  The report table 
can include all of the same statistics as the plot and can be re-ordered within the table. 

Table 2 shows an example of the report table. 

Example Report Table 

 Event/Total 
Hazard Ratio 
(95% CI)

Cox
 

Covariate 
Level 

P-values 
Hazard Ratio 
(95% CI)

1Cox
 

Covariate 
Level 

P-values
1
 

DIAGNOSIS      

ALL 24/38 1.78 (1.01-3.12) 0.0457
+
 1.87 (1.06-3.31) 0.0313

+
 

AML high risk 34/45 2.60 (1.55-4.38) 0.0003
+
 2.63 (1.56-4.43) 0.0003

+
 

AML low risk 25/54 Reference -- Reference -- 
Cox

Cox model; 
+
Wald Chi-Square test; 

1
Adjusted for gender 

 Table 2. The example report table shows a comparison of a univariate model and a multivariate model that is 
adjusted for gender. 

%newsurv( 

    /*Required Parameters*/ 

    data=bmt,time=ftime,cens=status, 

    /*Class Variable Options*/ 

    class=diagnosis, 

    /*Adjusting Factors*/ 

    classcov=gender, 

    /*Table Display Option*/ 

    tabledisplay=ev_n hr covpval hrmv covpvalmv, 

    /*Table Title/footnote Options*/ 

    tabletitle=Example Report Table,tablefootnote=^{super 1}Adjusted for gender, 

    /*Table Header Options*/ 

    tcovpvalmvheader=Covariate^nLevel^nP-values^{super 1}, 

thrmvheader=Hazard Ratio^n(95% CI)^{super 1}, 

    /*Output File Options*/ 

    plot=0,destination=rtf,outdoc=table2.rtf); 

The ^n strings in several of the parameters above are used to create line breaks.  The ^ is set as the ODS 
ESCAPECHAR within the macro for this purpose.  The ^{super 1} strings create a superscript 1 value that is used as 
footnote markers. 

4.1.4 Creating Multiple Plots within the Same Image 

Often times multiple plots are set next to each other in order to make a comparison or to save on the number of 
figures within a manuscript.  The NEWSURV macro allows multiple models to be run and plotted within the same 
image for this purpose.  Macro parameters can be set for each plot uniquely using the | symbol. 

Figure 3 is an example of creating multiple plots within one image 

 

Figure 3. This example demonstrates showing the same graph across two levels of a variable.  It can be 
clearly determined that the survival is different depending on the gender of the patient. 

%newsurv( 

    /*Required Parameters*/ 

    data=bmt,time=ftime,cens=status, 

    /*Multiple Model Options*/ 

    nmodels=2,rows=2, 

    /*Where Clause to Subset*/ 

    where=gender=1|gender=0, 

    /*Class Variable Options*/ 

    class=diagnosis, 

    /*Line/Symbol Options*/ 

    linesize=2pt,symbolsize=10pt,color=black blue red,pattern=1, 

    /*Axis Options*/ 

    xdivisor=30.44,ylabel=Percent Alive,xlabel=Months,xmax=100,xincrement=20, 

    /*Title Options*/ 

    title=Survival within Male Patients|Survival within Female Patients, 

/*Output File Options*/  

plotname=figure3,height=7in); 

4.2 NEWSURV MACRO NEW FEATURES 

4.2.1 Cumulative Incidence and Competing Risks 

The original NEWSURV macro could plot survival and 1-survival, but not true cumulative incidence.  The current 
version of the macro has adapted the SAS auto call macro CIF, but instead of doing the calculations with the IML 
procedure the NEWSURV macro uses the DATA step with array computations.  SAS version 9.4M3 currently can do 
these calculations with the LIFETEST procedure, but the DATA step computations are currently being kept to allow 
functionality back to SAS 9.2.  The Gray k-sample test for comparing curves is also available.  Hazard ratios based 
on Fine and Gray’s competing risks are computed using the PHREG procedure, but is only available in SAS 9.4M1 or 
later.   

Figure 4 is an example comparing 1-survival to cumulative incidence (Specifying death as the event) 

Figure 4. The difference between 1-survival and cumulative incidence is evident within this graph.  There are 
far less events in the second curve as the relapses are no longer considered events. 

Plot of Cumulative Incidence

%newsurv( 

/*Required Parameters*/ 

data=bmt,time=ftime,cens=status,method=km|cif,ev_vl=|1, 

/*Multiple Model Options*/ 

nmodels=2,rows=2, 

/*Summary table options*/ 

autoalign=bottomright|topright,display=legend ev_n median hr pval, 

/*Plot 1-S for first model*/ 

sreverse=1|0, 

/*Class Variable Options*/ 

class=diagnosis, 

/*Line/Symbol Options*/ 

linesize=2pt,symbolsize=10pt,color=black blue red,pattern=1, 

/*Axis Options*/ 

xdivisor=30.44,ylabel=Percent Dead,xlabel=Months,xmax=100,xincrement=20, 

/*Title Options*/ 

title=Plot of 1-S|Plot of Cumulative Incidence, 

/*Output File Options*/ 

plotname=figure4, height=7in); 

4.2.2 Reference Lines 

Reference lines add another level to visually comparing multiple survival curves.  The NEWSURV macro can draw 
reference lines on either the median time-to-event values or time-point event-free rates.  This allows those viewing 
the graph to quickly compare key differences of the curves without seeing the numbers. 

Figure 5 Adds reference lines to highlight the median time-to-event. 

Figure 5. The median survival of the AML lower-risk group is clearly much greater than the other two groups 
which implies a much higher efficacy of the treatment within the AML lower-risk group. 

00 2200 4400 6600 8800 110000

MMoonntthhss

00

1100

2200

3300

4400

5500

6600

7700

8800

9900

110000

PP
eerr

ccee
nn

tt  
AA

lliivv
ee

CensorLogrank P-value: 0.0010
Reference72.4 (23.1-NE)25/54

ALL
AML high risk
AML low risk

13.7 (6.4-NE) 1.78 (1.01-3.12)
6.0 (3.8-15.0) 2.60 (1.55-4.38)34/45

24/38
DIAGNOSIS Events/Total Median (95% CI) HR (95% CI)



The %NEWSURV Family of Macros: An Update on the Survival Plotting Macro %NEWSURV and an Introduction to Expansion Macros, 
continued 

8 

%newsurv( 

/*Required Parameters*/ 

data=bmt,time=ftime,cens=status, 

/*Class Variable Options*/ 

class=diagnosis, 

/*Reference Lines*/ 

reflines=medians,reflinemethod=drop,reflineaxis=both, 

/*Line/Symbol Options*/ 

linesize=2pt,symbolsize=10pt,color=black blue red,pattern=1, 

/*Axis Options*/ 

xdivisor=30.44,ylabel=Percent Alive,xlabel=Months,xmax=100,xincrement=20, 

/*Output File Options*/ 

svg=1,destination=pdf,outdoc=~/ibm/figure5.pdf,plotname=figure5,gpath=~/ibm/); 

4.2.3 Confidence Intervals 

Confidence intervals allow the user to quickly determine if two curves are statistically different from each other.  The 
NEWSURV macro creates 95 percent confidence intervals with a variety of methods, and can be drawn as 
transparent color bands, as dashed lines, or both.  When the confidence intervals of two curves do not overlap it is an 
indication that the curves are significantly different.  When a single curve is plotted the confidence intervals give an 
indication of the stability of the data.  A confidence interval that quickly expands throughout the curve indicates that 
there are not enough events currently in the data.  NEWSURV turns on confidence intervals by default for single 
curve plots and can be turned on for multi-curve plots. 

Figure 6 Adds confidence intervals in the form of transparent color bands. 

Figure 6. The confidence bands for the AML high-risk and AML low-risk groups do not overlap until the very 
end of the curves.  This gives the indication that the curves are significantly different from each other and 
could be the main driver of the significant p-value. 

5.0 EXPANSION MACRO NEWSURV_DATA 

The NEWSURV_DATA macro was created in order to be able to plot adjusted survival curves.  There are numerous 
methods available for creating adjusted survival curves, and none are the correct method for all situations.  Thus 
there was a need to have a macro that could create the high quality plots of NEWSURV, but with the flexibility to be 
able to plot any provided set of survival function data.  This macro was designed to create one graph at a time and 
does not have the functionality to create a report summary, but it can still create the plot summary table with a 
combination of statistics calculated from the survival function data and a possible input dataset. 

5.1 INPUT DATASET 

The NEWSURV_DATA macro requires one input dataset and allows a second one to be optional.  The required 
dataset only needs to have a time variable and a variable containing survival function estimates.  The macro also 
allows optional variable inputs for a class variable, lower confidence limits, upper confidence limits, and for patients-
at-risk numbers.  The class variable will be used to group the survival curves.  The lower and upper confidence limits 
are required when plotting the confidence limits, and are also used for the 95 percent confidence limits on the median 
time-to-event and time-point event-free rates.  The patients-at-risk variable contains numbers that coincide with the 
time variable, and these values are used to build the patients-at-risk table within the plot. 

The optional input dataset is used to build the plot summary table and can contain any number of variables.  If a 
CLASS variable is specified then it must also exist within this input dataset for merging later. 

5.2 STATISTICS AVAILABLE TO DISPLAY BY DEFAULT 

There are two statistics that the macro can calculate from the required input dataset: median time-to-event and time-
point event-free rates. 

5.2.1 Median Time-to-Event 

The median time-to-event is calculated in one of two ways depending on where the survival function begins at time 
zero.  If the survival function starts at one or 100 percent, then the median time-to-event is the first time where the 
survival function drops below 0.5 or 50 percent.  The lower bound of the 95 percent confidence interval is the time 
when the lower limit of the survival function first drops below 0.5 or 50 percent, and the upper bound of the 95 percent 
confidence interval is the time when the upper limit of the survival function first drops below 0.5 or 50 percent.  
Otherwise if the survival function starts at 0 then the median time-to-event is the time where the survival function first 
increases above 0.5 or 50 percent.  The lower bound of the 95 percent confidence interval is the time when the upper 
limit of the survival function first increases above 0.5 or 50 percent, and the upper bound of the 95 percent confidence 
interval is the time when the lower limit of the survival function first increases above 0.5 or 50 percent. 

5.2.2 Time-point Event-Free Rates 

One or more time-point event-free rates can be calculated from the inputted dataset.  The time-point estimate is the 
value of the survival function at the time that is closest to the specified time-point and is either before or equal to the 
time-point.  The 95 percent confidence interval is calculated in two ways depending on the survival function at time 
zero.  If the survival function begins at one or 100 percent, then the lower limit and upper limit make up the lower and 
upper bounds of the 95 percent confidence interval respectively.  Otherwise if the survival function begins at 0 then 
the upper limit and lower limit make up the lower and upper bounds of the 95 percent confidence interval respectively.   

5.3 ADDING STATISTICS TO PLOT SUMMARY TABLE 

Below is a complete example of creating and using an optional summary table for the plot. 

5.3.1 Computing the Survival Function Estimates and Number of Patients/Events 

The LIFETEST procedure creates the data set _SURV containing times and survival function estimates based on 
ODS table SURVIVALPLOT.  This data set is the one LIFETEST uses to generate its own plot and is more complete 
than using the OUTS option.  The _SUM dataset is created from the CENSOREDSUMMARY ODS table and contains 
the number of patients and events within each level of DIAGNOSIS. 

proc lifetest data=bmt plot=(survival); 

    strata diagnosis; 

    time ftime*status(0); 

    ods output survivalplot=_surv censoredsummary=_sum; 

run; 

Table 3 shows the _SUM data set 

control_var Stratum DIAGNOSIS Total Failed Censored PctCens 

  1 ALL 38 24 14 36.84 

  2 AML high risk 45 34 11 24.44 

  3 AML low risk 54 25 29 53.70 

- T _ 137 83 54 39.42 

 Table 3. The _SUM data set has variables for the levels of the stratum group (STRATUM) and DIAGNOSIS.  
There are also variables for the number of patients (TOTAL), number of events (FAILED), number of censors 
(CENSORED) and percent censored (PCTCENS).  The final variable, CONTROL_VAR, indicates which rows 
are part of the stratum groups (missing) and which is the total column (not missing). 

Table 4 shows the first several rows of the _SURV data set 

Time Survival AtRisk Event Censored Stratum StratumNum 

0 1.00000 38 0   ALL 1 

1 .97368 38 1   ALL 1 

55 .94737 37 1   ALL 1 

74 .92105 36 1   ALL 1 

86 .89474 35 1   ALL 1 

104 .86842 34 1   ALL 1 

107 .84211 33 1   ALL 1 

109 .81579 32 1   ALL 1 

110 .78947 31 1   ALL 1 

122 .73684 30 2   ALL 1 

129 .71053 28 1   ALL 1 

172 .68421 27 1   ALL 1 

 Table 4. The _SURV data set has variables for time (TIME), survival function estimates at events (SURVIVAL), 
number of patients at risk (ATRISK), number of events occurring at the current time (EVENT), survival 
function estimate at censored patients (CENSORED), what the current stratum group is (STRATUM) and a 
numeric representation of the current stratum group (STRATUMNUM).  

5.3.2 Computing the Hazard Ratio and P-values 

The PHREG procedure creates the hazard ratios and p-values of the univariate model based on the DIAGNOSIS 
variable and saves these values into the _PARM dataset based on the PARAMETERESTIMATES ODS table. 

proc phreg data=bmt; 

    class diagnosis (ref='ALL'); 

    model ftime*status(0)=diagnosis / rl;  

    ods output parameterestimates=_parm; 

run; 

 

Table 5 shows the _PARM data set 

Parameter ClassVal0 DF Estimate StdErr ChiSq ProbChiSq HazardRatio HRLowerCL HRUpperCL 

DIAGNOSIS 

AML high 

risk 1 0.38262 0.26738 2.0478 0.1524 1.466 0.868 2.476 

DIAGNOSIS 

AML low 

risk 1 -0.57418 0.28730 3.9942 0.0457 0.563 0.321 0.989 

 Table 5. The _PARM data set contains variables for the current covariate (PARAMETER), level of the 
covariate if in the CLASS statement (CLASSVAL0), degrees of freedom (DF), model coefficient (ESTIMATE), 
standard error (STDERR), chi-square test value (CHISQ), Wald chi-square p-value (PROBCHISQ), hazard ratio 
(HAZARDRATIO), hazard ratio 95 percent lower limit (HRLOWERCL), and hazard ratio 95 percent upper limit 
(HRUPPERCL). 



The %NEWSURV Family of Macros: An Update on the Survival Plotting Macro %NEWSURV and an Introduction to Expansion Macros, 
continued 

 

11 

5.3.3 Preparing to Merge _SUM and _PARM Data Sets 

The following code creates a variable named CLASSVAL0 to match the variable within the _PARM data set created 
in the previous step and subsets to just the levels of the DIAGNOSIS Variable.  

data _sum; 

    set _sum; 

    classval0=vvalue(diagnosis); 

    where missing(control_var); 

    keep total failed classval0; 

run; 

 

Table 6 shows the updated _SUM table 

Total Failed classval0 

38 24 ALL 

45 34 AML high risk 

54 25 AML low risk 

Table 6. The VVALUE function returns the formatted value of the inputted variable as a character variable 
which matches how the CLASSVAL0 variable is made in the _PARM data set.  Subsetting the data set to 
where CONTROL_VAR is not missing removes the total row from the data set.  

5.3.3 Merging the _SUM and _PARM Data Sets and Creating Report Variables 

The following merges the _SUM and _PARM datasets based on the CLASSVAL0 variable mentioned earlier.  The 
code creates two new variables EV_N and HR that concatenate values together into a character variable to show in 
the plot table.   

 EV_N combines number of events and number of patients with a / between.   

 HR combines the hazard ratio with its lower and upper limits in a ##.## (##.##-##.##) format.   

The CLASSVAL0 variable is renamed to STRATUM to match the variable in the _SURV dataset created earlier.  This 
is important for the NEWSURV_DATA macro call later since STRATUM will be the CLASS variable.  _SUM contains 
all three levels of the DIAGNOSIS variable, but _PARM does not as it does not display the reference group.  Thus the 
code adds the value REFERENCE to the value of CLASSVAL0 that does not exist in _PARM. 

data _table; 

    merge _sum _parm (in=b); 

    by classval0; 

    length ev_n hr $50.; 

    ev_n=strip(put(failed,12.0))||'/'||strip(put(total,12.0)); 

    if b then hr=strip(put(hazardratio,12.2))||' ('|| 

      strip(put(hrlowercl,12.2))||'-'||strip(put(hruppercl,12.2))||')'; 

    else hr='Reference'; 

    rename classval0=stratum; 

    keep classval0 ev_n hr probchisq; 

    label hr='Hazard Ratio (95% CI)' probchisq='P-value' ev_n='Events/Total'; 

run; 

 

Table 7 shows the _TABLE data set 

stratum ProbChiSq ev_n hr 

ALL   24/38 Reference 

AML high risk 0.1524 34/45 1.47 (0.87-2.48) 

AML low risk 0.0457 25/54 0.56 (0.32-0.99) 

Table 7.  The final plot summary data set _TABLE will be input into the NEWSURV_DATA macro.  The 
STRATUM variable is the same as the CLASS variable in the macro call.  The PROBCHISQ is the p-value from 
the _PARM data set.  EV_N is the combined TOTAL and FAILED columns from the _SUM data set.  HR is the 
combined HAZARDRATIO, HRLOWERCL and HRUPPERCL variables from the _PARM data set.  

5.3.4 Calling the NEWSURV_DATA Macro 

The following calls the macro.  _SURV is used for the time (TIME), survival function (SURVIVAL), and class 
(STRATUM) variables.  _TABLE is supplied with the class (STRATUM) and statistics to be displayed (EV_N, HR, and 
PROBCHISQ.     

%newsurv_data( 

    /*Required Parameters*/ 

    data=_surv,time=time,survival=survival,DISPLAY=legend var2 var3 var1, 

    /*Optional Dataset*/ 

    SUMMARYDATA=_table, 

    /*Class Variable Options*/ 

    class=stratum, 

    /*Line/Symbol Options*/ 

    linesize=2pt,symbolsize=10pt,color=black blue red,pattern=1, 

    /*Axis Options*/ 

    xdivisor=30.44,ylabel=Percent Alive,xlabel=Months,xmax=100,xincrement=20, 

    /*Output File Options*/ 

    plotname=figure7) 

 

 

Figure 7 Plot created from NEWSURV_DATA example. 

 

Figure 7. The values of the legend (STRATUM), events and total (EV_N), hazard ratios (HR), and p-values 
(PROBCHISQ) can be seen within the plot in the upper right hand corner. 

6.0 EXPANSION MACROS NEWSURV_ADJ_INVWTS AND NEWSURV_ADJ_DIRECT 

6.1 BACKGROUND ON ADJUSTED SURVIVAL CURVES 

There are many different methods that can be used to do adjusted survival that are all very situational to the dataset 
and patient population, and because of this and discussions with statisticians at Mayo Clinic it was decided to not add 
limited methods to the NEWSURV macro itself in order to not imply that they would be the only viable methods.  
Instead two popular methods were incorporated into the following two macros: NEWSURV_ADJ_INVWTS and 
NEWSURV_ADJ_DIRECT.  The NEWSURV_ADJ_INVWTS macro uses the inverse weights method

4 
and the

NEWSURV_ADJ_DIRECT uses the direct adjustment method
4
.  The options to use either macro are the same, but

the Kaplan-Meier curves will be adjusted differently depending on the macro used.  There will be preferences by 
statisticians as to the appropriate method to be used. 

6.2 METHODS BEHIND NEWSURV_ADJ_INVWTS 

6.2.1 Description of the Method 

The inverse weights method involves the following steps to calculate weights for an adjusted Kaplan-Meier curve: 

1. Run one logistic regression model where the dependent variable is a binary version of the CLASS variable
and the independent covariates are the adjusting factors and the stratification factors.  If the CLASS variable
has more than two levels, then the binary variable is programmed the following way:

a. CLASS has three levels A, B, and C

b. Model 1 will run with A as the event and B/C as the non-event
c. Model 2 will run with B as the event and A/C as the non-event
d. Model 3 will run with C as the event and A/B as the non-event

2. Each of these models will output a set of weights.  These weights are then reweighted based upon the
frequency of the CLASS variable

a. Model 1 will reweight the weights of A with the following formula: (NA/SUM(NA,NB,NC))/WeightA
b. Model 2 will reweight the weights of B with the following formula: (NB/SUM(NA,NB,NC))/WeightB
c. Model 3 will reweight the weights of C with the following formula: (NC/SUM(NA,NB,NC))/WeightC

3. The final weight for each level of the CLASS variable will be the reweighted weight from the model where
that respective class level was the event of the logistic model

a. CLASS level A uses the weights from step 2a
b. CLASS level B uses the weights from step 2b
c. CLASS level C uses the weights from step 2c

4. These weights are then supplied with the WEIGHTS statement within the LIFETEST procedure to adjust the
survival curves.

6.2.2 Example Macro Call 

The subsequent sections have code that can be viewed within the log for the following macro call while the MPRINT 
option is enabled to view the actual programming methods the macro NEWSURV_ADJ_INVWTS is using to create 
the weights. 

%newsurv_adj_invwts( 

/*Required Parameters*/ 

data=bmt,time=ftime,cens=status, 

/*Class Variable Options*/ 

class=diagnosis, 

/*Adjusting Factors*/ 

classcov=gender, 

/*Line/Symbol Options*/ 

linesize=2pt,symbolsize=10pt,color=black blue red, 

/*Axis Options*/ 

xdivisor=30.44,ylabel=Percent Alive,xlabel=Months,xmax=100,xincrement=20, 

/*Output File Options*/ 

plotname=figure8);  

6.2.3 Preparing for the LOGISTIC Procedure Call 

The following DATA step is done in order to allow the macro to only have to call the LOGISTIC procedure one time 
even though it is running one model per level of the CLASS variable.  This saves running time especially on large 
data sets. The DATA step duplicates the inputted data set one time for each level of the CLASS variable.  Each 
time 

the data set is duplicated one of the levels of the CLASS variable is marked as the event for the logistic model and 
the others are marked as non-events. The array is created from macro variables based upon the CLASS variable.  
The CLASS variable itself has been renamed to _CLASS_ within the macro. 

data _adjwts_prep; 

set _tempdsn1; 

array _class_lvls_ { 3} $100. ( "ALL" , "AML high risk" , "AML low risk"); 

_order_=_n_; 

do i = 1 to 3; 

_set_=i; 

if _class_=_class_lvls_(i) then _adjevent_=1; 

else _adjevent_=0; 

output; 

end; 

drop i _class_lvls_:; 

 run; 

 /**Orders the data set for the following LOGISTIC procedure.**/ 

 proc sort data=_adjwts_prep; 

by _set_; 

 run; 

6.2.4 Running the LOGISTIC Procedure 

The LOGISTIC procedure is run using the _SET_ variable created in 3.2.3 to run three logistic models at the same 
time.  The weights are then saved into the dataset _ADJWTS using the OUTPUT statement and the weights are 
named _ADJWTS_ within the dataset.  The _ADJWTS retains all of the original variables from the _ADJWTS_PREP 
dataset and adds the newly created _ADJWTS_ variable. The _CLASSCOV_1 variable is the GENDER variable 
from the macro call renamed within the macro.   

 proc logistic data=_adjwts_prep; 

by _set_; 

class _classcov_1; 

model _adjevent_ (event='1') = _classcov_1 ; 

output out=_adjwts prob=_adjwts_; 

 run; 

 /**Sorts by the ORDER variable which is the original order of the data**/ 

 proc sort data=_adjwts; 

by _order_ _set_; 

 run; 

6.2.5 Reweighting and Assigning the Final Weights 

The _ADJWTS data set is merged into itself one time per level of the CLASS variable, but each merging only keeps 
one model and renames the _ADJWTS_ variables to be used within an array.  This is essentially transposing the 
_ADJWTS data set.  The final weight is stored into the _ADJWT_ variable created within this DATA step based on 
the level of the CLASS variable using the formulas in 3.2.1.  The array with the counts (_CLASS_N_) is created by 
the macro. 

 data _adjwts; 

merge _adjwts (where=(_set_=1) rename=(_adjwts_=_adjwts_1)) 

_adjwts (where=(_set_=2) rename=(_adjwts_=_adjwts_2) 

keep=_set_ _order_ _adjwts_)  

   _adjwts (where=(_set_=3) rename=(_adjwts_=_adjwts_3) 

keep=_set_ _order_ _adjwts_); 

by _order_; 

array _class_lvls_ { 3} $100. ( "ALL" , "AML high risk" , "AML low risk"); 

array _class_n_ { 3} ( 38 , 45 , 54); 

array _adjwts_ { 3}; 

do i = 1 to 3; 

if _class_=_class_lvls_(i) then _adjwt_=(_class_n_(i)/sum(of 

_class_n_(*)))/_adjwts_(i); 

end; 

drop _set_ _order_ _adjevent_ _level_ i _class_lvls_: _class_n_: _adjwts_:; 

 run; 

6.2.6 Creating the Weighted Survival Curves 

The _ADJWTS dataset is then supplied to the LIFETEST procedure and the _ADJWT_ variable is supplied within the 
WEIGHTS statement in order to adjust the curves.  The rest of the LIFETEST code is necessary for the macro to run. 

 ods graphics on; 

 proc lifetest data=_adjwts plot=(survival(cl)) conftype=LOG; 

strata _class_ / logrank; 

time _time_ * _cens_(0); 

weight _adjwt_; 

ods output censoredsummary=_sum_adj 

  quartiles=_quart_adj (where=(percent=50)) s 

  homtests=_ltest_adj  

survivalplot=_surv_adj (where=(event>.) rename=( stratum=cl1 

survival=s1 time=t1 censored=c1 sdf_lcl=lcl1 sdf_ucl=ucl1) 

keep=survival time censored event sdf_lcl sdf_ucl stratum  

stratumnum); 

 run; 

Figure 8 shows the unadjusted Kaplan-Meier curves and the Kaplan-Meier curves adjusted for gender with the 
inverse weights methods. 

Figure 8. The dashed lines represent the adjusted Kaplan-Meier curves.  The adjusted curves are essentially 
the same as the unadjusted curves in this case, and this is reflected in the statistical summary table where 
the medians, hazard ratios and p-values are nearly identical. 

6.3 METHODS BEHIND NEWSURV_ADJ_DIRECT 

6.3.1 Description of the Method  

The direct adjustment method involves the following steps to create adjusted Kaplan-Meier curves: 

1. Sets up data set to be used with the COVARIATES option of the BASELINE statement of the PHREG
procedure

a. This dataset provides the population that the PHREG procedure uses to adjust the prediction
curves

b. The dataset is duplicated one time for each level of the CLASS variable and then set together to

make one data set
c. Each duplication of the data set has the entire CLASS variable’s value set to one value of the

CLASS variable
i. CLASS variable has values A, B, and C
ii. Duplication 1 will set CLASS to all be A
iii. Duplication 2 will set CLASS to all be B
iv. Duplication 3 will set CLASS to all be C

d. All other variables are left alone
2. The original data set is supplied to the PHREG procedure and the new data set containing the duplications

is supplied to the BASELINE statement with the COVARIATES and DIRADJ options.
a. The CLASS variable is placed within the STRATA statement.  This combined with the BASELINE

statement forces the PHREG procedure to produce one adjusted prediction curve per level of the
CLASS variable.

b. The data set containing the adjusted survival curves are output with the OUT option within the
BASELINE statement

6.3.2 Example Macro Call 

The subsequent sections have code that can be viewed within the log for the following macro call while the MPRINT 
option is enabled to view the actual programming methods the macro NEWSURV_ADJ_DIRECT is using to create 
the adjusted curves. 

%newsurv_adj_direct( 

/*Required Parameters*/ 

data=bmt,time=ftime,cens=status, 

/*Class Variable Options*/ 

class=diagnosis, 

/*Adjusting Factors*/ 

classcov=gender, 

/*Line/Symbol Options*/ 

linesize=2pt,symbolsize=10pt,color=black blue red, 

/*Axis Options*/ 

xdivisor=30.44,ylabel=Percent Alive,xlabel=Months,xmax=100,xincrement=20, 

/*Output File Options*/ 

plotname=figure9); 

6.3.3 Creating the COVARIATES Data Set for the BASELINE Statement 

The following DATA step sets the inputted data set one time for each level of the CLASS variable.  Each 
duplication of the data set sets the CLASS variable (renamed _CLASS_ by the macro) to be equal to one of the 
values of CLASS. 
data _covs; 

set _tempdsn1 (in=a1) _tempdsn1 (in=a2) _tempdsn1 (in=a3); 

if a1 then _class_="ALL"; 

else if a2 then _class_="AML high risk"; 

else if a3 then _class_="AML low risk"; 

 run; 

6.3.4 Running the PHREG Procedure to Create the Adjusted Kaplan-Meier Curves 

The previously created _COVS data set to the COVARIATES option within the BASELINE statement.  The 
DIRADJ option tells SAS to use the direct adjustment method to produce adjusted prediction curves based on the 
population provided in the BASELINE statement.  The CLASS (renamed _CLASS_) variable is put into the 
STRATA statement in order to group the prediction curves by the CLASS variable.  Only the predicted survival 

curves are taken from this PHREG procedure call; the hazard ratios and other statistics are ignored. 
 ods graphics on; 

 proc phreg data=_tempdsn1 plots(cl) =survival ; 

strata _class_; 

class _classcov_1; 

model _time_*_cens_(0)= _classcov_1 / rl; 

baseline covariates=_covs out=_surv_adj (rename=(_time_=t1 _class_=cl1)) 

survival=s1 lower=lcl1 upper=ucl1/ diradj cltype=LOG; 

where missing(_class_)=0 and missing(_classcov_1)=0; 

 run; 

Figure 9 shows the unadjusted Kaplan-Meier curves and the Kaplan-Meier curves adjusted for gender under the 
direct adjustment method. 

Figure 9. The curves are adjusted very similarly to figure 8 with the primary difference being in the AML low 
risk group is adjusted above the unadjusted group instead of below it in figure 8.  The direct adjustment 
method will often adjust the curves more drastically than the inverse weights method. 

6.4 DIFFERENCES SPECIFIC TO NEWSURV_ADJ MACROS 

There are two primary differences between the original NEWSURV and the new NEWSURV_ADJ macros.  The first 
is in the plot summary table.  Previously adjusted statistics such as hazard ratios were listed in new columns where in 
the adjusted macros the adjusted statistics are displayed in new rows.  When both adjusted and unadjusted curves 
are plotted the term (Adjusted) is appended to the rows that designate the adjusted curves.  The second difference 
comes in the report table.  In the original NEWSURV adjusted statistics could be requested individually whether or 
not the unadjusted statistics were displayed.  In the new adjusted macros whichever statistics are requested to be 
displayed are shown in both unadjusted and adjusted versions. 

Table 3 shows an example of the report table created by the adjusted survival macros 

Unadjusted Survival Adjusted Survival 

Event/Total 
Hazard Ratio 
(95% CI)

Cox

Covariate 
Level 

P-values Event/Total 
Hazard Ratio 
(95% CI)

Cox

Covariate 
Level 

P-values

DIAGNOSIS 

ALL 24/38 1.78 (1.01-3.12) 0.0457
+

24/38 1.87 (1.06-3.31) 0.0313
+

AML high risk 34/45 2.60 (1.55-4.38) 0.0003
+

34/45 2.63 (1.56-4.43) 0.0003
+

AML low risk 25/54 Reference -- 25/54 Reference -- 
Cox

Cox model; 
+
Wald Chi-Square test;

Table 3. The unadjusted and adjusted survival statistics each have their own sections with subtitle. 

7.0 REAL WORLD EXAMPLE 

Below is an example of adjusted curves from oncology clinical trial data that spurred the creation of the 
NEWSURV_ADJ macros.  The univariate model involving the biomarker showed a significant difference between 
level 1 and level 2, but after adjusting for other factors the hazard ratios reversed direction and significance.  Figure 
10 graphically displays the drastic difference from regular Kaplan-Meier curves and adjusted Kaplan-Meier curves.  
This illustrates the care that is required when reporting univariate and multivariate model results but only unadjusted 
survival curves. 

Figure 10 shows the unadjusted vs adjusted curves from real oncology trial data. 

Figure 10. Adjusting the Cox proportional hazards model by certain covariates caused the hazard ratio for 
the biomarker of interest to completely flip such that level 1 of the biomarker went from having a worse 
outcome to having a better outcome.  Plotting the adjusted curves reflects that dramatic shift in just how 
much the level 1 curve of the biomarker shifted vertically upwards. 

8.0 CONCLUSION  

The NEWSURV family of macros are a powerful set of tools for survival analysis, and they are ever evolving to meet 
the needs of the community of users. There are quick and yet simple improvements with quick turnaround as well as 
complicated topics that require research and collaboration.   Adjusted Kaplan-Meier curves are an intriguing area of 
study and is worthy of being the focus of the SAS programming community.  The next focus of the NEWSURV 
macros will be the Aalen-Johansen method of multi-state survival analysis and exploring patients being able to 
transfer from one state to another over time. 

10.0 RECOMMENDED READING  

 http://www.sascommunity.org/wiki/Kaplan-Meier_Survival_Plotting_Macro_%25NEWSURV 

 Website with prior papers, presentations, and macro versions for download 

 Description of the macro features 

 Example code for using the macro 

 SAS 9.3 Graph Template Language Reference, Third Edition 

 SAS 9 PROC TEMPLATE Styles Tip Sheet 

12.0 APPENDIX I  

The following code can be run to create the BMT data set used in examples. 

proc format; 

   value grpLabel 1='ALL' 2='AML low risk' 3='AML high risk'; 

   value gender 1='Male' 0='Female'; 

run; 

 

data BMT; 

        input DIAGNOSIS Ftime Status Gender@@; 

        label Ftime="Days"; 

        format Diagnosis grpLabel. Gender gender.; 

https://cran.r-project.org/web/packages/survival/vignettes/adjcurve.pdf
https://cran.r-project.org/web/packages/survival/vignettes/adjcurve.pdf
http://www.sascommunity.org/wiki/Kaplan-Meier_Survival_Plotting_Macro_%25NEWSURV

datalines; 

1       2081       0       1       1       1602    0       1 

1       1496       0       1       1       1462    0       0 

1       1433       0       1       1       1377    0       1 

1       1330       0       1       1       996     0       1 

1       226        0       0       1       1199    0       1 

1       1111       0       1       1       530     0       1 

1       1182       0       0       1       1167    0       0 

1       418        2       1       1       383     1       1 

1       276        2       0       1       104     1       1 

1       609        1       1       1       172     2       0 

1       487        2       1       1       662     1       1 

1       194        2       0       1       230     1       0 

1       526        2       1       1       122     2       1 

1       129        1       0       1       74      1       1 

1       122        1       0       1       86      2       1 

1       466        2       1       1       192     1       1 

1       109        1       1       1       55      1       0 

1       1          2       1       1       107     2       1 

1       110        1       0       1       332     2       1 

2       2569       0       1       2       2506    0       1 

2       2409       0       1       2       2218    0       1 

2       1857       0       0       2       1829    0       1 

2       1562       0       1       2       1470    0       1 

2       1363       0       1       2       1030    0       0 

2       860        0       0       2       1258    0       0 

2       2246       0       0       2       1870    0       0 

2       1799       0       1       2       1709    0       0 

2       1674       0       1       2       1568    0       1 

2       1527       0       0       2       1324    0       1 

2       957        0       1       2       932     0       0 

2       847        0       1       2       848     0       1 

2       1850       0       0       2       1843    0       0 

2       1535       0       0       2       1447    0       0 

2       1384       0       0       2       414     2       1 

2       2204       2       0       2       1063    2       1 

2       481        2       1       2       105     2       1 

2       641        2       1       2       390     2       1 

2       288        2       1       2       421     1       1 

2       79         2       0       2       748     1       1 

2       486        1       0       2       48      2       0 

2       272        1       0       2       1074    2       1 

2       381        1       0       2       10      2       1 

2       53         2       0       2       80      2       0 

2       35         2       0       2       248     1       1 

2       704        2       0       2       211     1       1 

2       219        1       1       2       606     1       1 

3       2640       0       1       3       2430    0       1 

3       2252       0       1       3       2140    0       1 

3       2133       0       0       3       1238    0       1 

3       1631       0       1       3       2024    0       0 

3       1345       0       1       3       1136    0       1 

3       845        0       0       3       422     1       0 

3       162        2       1       3       84      1       0 

3       100        1       1       3       2       2       1 

3       47         1       1       3       242     1       1 

3       456        1       1       3       268     1       0 

3       318        2       0       3       32      1       1 

3       467        1       0       3       47      1       1 

3       390        1       1       3       183     2       0 

3       105        2       1       3       115     1       0 

3       164        2       0       3       93      1       0 

3       120        1       0       3       80      2       1 

3       677        2       1       3       64      1       0 

3       168        2       0       3       74      2       0 

3       16         2       0       3       157     1       0 

3       625        1       0       3       48      1       0 

3       273        1       1       3       63      2       1 

3       76         1       1       3       113     1       0 

3       363        2       1 

; 

run; 

 




